

CSc 372 — Comparative Programming Languages

24 : Prolog — Negation

Christian Collberg
Department of Computer Science
University of Arizona
collberg@gmail.com

Copyright © 2011 Christian Collberg

October 25, 2011

1

The Cut

2 Cuts & Negation

The cut (!) is used to affect Prolog's backtracking. It can be used to

- reduce the search space (save time).
- tell Prolog that a goal is deterministic (has only one solution) (save space).
- construct a (weak form of) negation.
- construct `if_then_else` and `once` predicates.

3 Cuts & Negation

- The cut reduces the flexibility of clauses, and destroys their logical structure.
- Use cut as a last resort.
- Reordering clauses can sometimes achieve the desired effect, without the use of the cut.
- If you are convinced that you have to use a cut, try using `if_then_else`, `once`, or `not` instead.

4

The Cut

5 The Cut


The cut succeeds and commits Prolog to all the choices made since the parent goal was called.

_____ Cut does two things: _____


commit: Don't consider any later clauses for this goal.

prune: Throw away alternative solutions to the left of the cut.

6 The Cut



7 The Cut




8

The Boxflow Model

9


The Boxflow Model

10 The Boxflow Model



11 The Boxflow Model


$a(X) :- b(X), c(X).$
 $a(X) :- d(X).$



12 The Cut

$a :- b, !, c.$
 $a :- d.$

if b then c
 else d



13

Classifying Cuts

14

Classifying Cuts

15 Classifying Cuts

grue No effect on logic, improves efficiency.

green Prune away

- irrelevant proofs
- proofs which are bound to fail

blue Prune away

- proofs a smart Prolog implementation would not try, but a dumb one might.

red Remove unwanted logical solutions.

16

Green Cuts

17 Green Cuts – Merge


Produce an ordered list of integers from two ordered lists of integers.

```
merge([X|Xs], [Y|Ys], [X|Zs]) :-
  X < Y, merge(Xs, [Y|Ys], Zs).
merge([X|Xs], [Y|Ys], [X,Y|Zs]) :-
  X = Y, merge(Xs, Ys, Zs).
merge([X|Xs], [Y|Ys], [Y|Zs]) :-
  X > Y, merge([X|Xs], Ys, Zs).
```

```
merge(Xs, [], Xs).
merge([], Ys, Ys).
```

```
?- merge([1,4], [3,7], L).
L = [1,3,4,7]
```

18 Green Cuts – Merge



19 Green Cuts

- Still, there is no way for Prolog to know that the clauses are mutually exclusive, unless we tell it so. Therefore, Prolog must keep all choice-points (points to which Prolog might backtrack should there be a failure) around, which is a waste of space.

- If we insert cuts after each test we will tell Prolog that the procedure is deterministic, i.e. that once one test succeeds, there is no way any other test can succeed. Prolog therefore does not need to keep any choice-points around.

20 Green Cuts – Merge

```
merge([X|Xs], [Y|Ys], [X|Zs]) :-
  X < Y, !,
  merge(Xs, [Y|Ys], Zs).
```


```
merge([X|Xs], [Y|Ys], [X,Y|Zs]) :-
  X = Y, !,
  merge(Xs, Ys, Zs).
```

```
merge([X|Xs], [Y|Ys], [Y|Zs]) :-
  X > Y, !,
  merge([X|Xs], Ys, Zs).
```

```
merge(Xs, [], Xs) :- !.
```

```
merge([], Ys, Ys) :- !.
```

21 Green Cuts – Merge



22

Red Cuts

23 Red Cuts – Abs

```
abs1(X, X) :- X >= 0.
```

```
abs1(X, Y) :- Y is -X.
```

```
?- abs1(-6, X).
```

```
  X = 6 ;
```

```
?- abs1(6, X).
```

```
  X = 6 ;
```

```
  X = -6 ;
```

```
abs2(X, X) :- X >= 0, !.
```

```
abs2(X, Y) :- Y is -X.
```

```
?- abs2(-6, X).
```

```
  X = 6 ;
```

```
?- abs2(6, X).
   X = 6 ;
```

24 Red Cuts – Abs

```
abs3(X, X) :- X >= 0.
abs3(X, Y) :- X < 0,
              Y is -X.
```

```
?- abs3(-6, X).
   X = 6 ;
   no
```

```
?- abs3(6, X).
   X = 6 ;
   no
```

25 Red Cuts – Intersection


Find the intersection of two lists A & B, i.e. all elements of A which are also in B.

```
intersect([H|T], L, [H|U]) :-
  member(H, L),
  intersect(T, L, U).
intersect([_|T], L, U) :-
  intersect(T, L, U).
intersect(_,_ , []).
```


26 Red Cuts – Intersection

```
?- intersect([3,2,1],[1,2], L).
   L = [2,1] ;
   L = [2] ;
   L = [2] ;
   L = [1] ;
   L = [] ;
   L = [] ;
   L = [] ;
   L = [] ;
   no
```


27 Red Cuts – Intersection



28 Red Cuts – Intersection



29 Red Cuts – Intersection




30 Red Cuts – Intersection

```
intersect([H|T], L, [H|U]) :-
    member(H, L),
    intersect(T, L, U).
intersect([_|T], L, U) :-
    intersect(T, L, U).
intersect(_,-, []).
```

```
intersect1([H|T], L, [H|U]) :-
    member(H, L), !,
    intersect1(T, L, U).
intersect1([_|T], L, U) :-
    !, intersect1(T, L, U).
intersect1(_,-, []).
```


31 Red Cuts – Intersection



32

Blue Cuts

33 Blue Cuts

First clause indexing will select the right clause in **constant** time:

```

clause(x(5), ...) :- ...
clause(y(5), ...) :- ...
clause(x(5, f), ...) :- ...
?- clause(x(C, f),...).

```

First clause indexing will select the right clause in **linear** time:

```

clause(W, x(5), ...) :- ...
clause(W, y(5), ...) :- ...
clause(W, x(5, f), ...) :- ...
?- clause(a, x(C, f),...).

```

34 Blue Cuts

```

capital(britain, london).
capital(sweden, stockholm).
capital(nz, wellington).
?- capital(sweden, X).
   X = stockholm
?- capital(X, stockholm).
   X = sweden

```

```

capital1(britain, london) :- !.
capital1(sweden, stockholm) :- !.

```

```
capital1(nz, wellington) :- !.
?- capital1(sweden, X).
   X = stockholm
?- capital1(X, stockholm).
   X = sweden
```

35

Once

36 Red Cuts – Once

```
member(H, [H|_]).
member(I, [_|T]) :- member(I, T).

?- member(1,[1,1]), write('x'), fail.
   xx

mem1(H, [H|_]) :- !.
mem1(I, [_|T]) :- mem1(I, T).
?- mem1(1, [1,1]), write('x'), fail.
   x
once(G) :- call(G), !.
one_mem(X, L) :- once(mem(X, L)).
?- one_mem(1,[1,1]), write('x'),fail.
   x
```

37 Red Cuts – Once

Red cuts prune away logical solutions. A clause with a red cut has no logical reading.

```
?- member(X, [1,2]).
   X = 1 ;
   X = 2 ;
   no

?- one_mem(X, [1,2]).
   X = 1 ;
   no
```

38

Cut & Fail & IF-THEN-ELSE

39 Red Cuts – Abs

```
abs2(X, X) :- X >= 0, !.
abs2(X, Y) :- Y is -X.
```

```
if_then_else(P,Q,R):-call(P),!,Q.  
if_then_else(P,Q,R):-R.
```

```
abs4(X, Y) :- if_then_else(X >= 0,  
                          Y=X, Y is -X).
```

```
?- abs4(-6, X).
```

```
    X = 6 ;  
    no
```

```
?- abs4(6, X).
```

```
    X = 6 ;  
    no
```

40 IF-THEN-ELSE

```
intersect([H|T], L, [H|U]) :-  
    member(H, L), !, intersect(T, L, U).  
intersect(_|T, L, U) :-  
    !, intersect(T, L, U).  
intersect(_,_ , []).
```

IF $H \in L$ **THEN**

compute the inters. of T and L,
let H be in the resulting list.

ELSEIF the list $\neq []$ **THEN**

let the resulting list be the
intersection of T and L.

ELSE

let the resulting list be [].

ENDIF

41 IF-THEN-ELSE

```
if_then_else(P,Q,R) :- call(P), !, Q.  
if_then_else(P,Q,R) :- R.
```

```
intersect2([X|T], L, W) :-  
    if_then_else(member(X, L),  
                (intersect2(T, L, U), W=[X|U]),  
                if_then_else(T \= [],  
                              intersect2(T, L, W),  
                              W = [])).
```

42

Negation

43

Negation

44 Open vs. Closed World

How should we handle *negative information*?

_____ Open World Assumption: _____

If a clause P is not currently asserted then P is neither true nor false.

_____ Closed World Assumption: _____

If a clause P is not currently asserted then the negation of P is currently asserted.

45 Open vs. Closed World

```
striker(dahlin).  
striker(thern).  
striker(andersson).
```

_____ Open World Assumption: _____

Dahlin, Thern, and Andersson are strikers, but there may be others we don't know about.

_____ Closed World Assumption: _____

X is a striker *if and only if* X is one of Dahlin, Thern, and Andersson.

46 Negation in Prolog

- Prolog makes the closed world assumption.
- Anything that I do not know and cannot deduce is not true.
- Prolog's version of negation is *negation as failure*.
- `not(G)` means that G *is not satisfiable as a Prolog goal*.

```
(1) not(G) :- call(G),!,fail.  
(2) not(G).
```

```
?- not(member(5, [1,3,5])).  
no  
?- not(member(5, [1,3,4])).  
yes
```

47 Prolog Execution – Not


- Some Prolog implementations don't define `not` at all. We then have to give our own implementation:

```
(1) not(G) :- call(G),!,fail.  
(2) not(G).
```

- Some implementations define `not` as
 - the operator `not`;
 - the operator `\+`;
 - the predicate `not(Goal)`.

`gprolog` uses `\+`.

48 Prolog Execution – Not



49 Negation Example – Disjoint

Do the lists `X` & `Y` **not** have any elements in common?

```
disjoint(X, Y) :-  
    not(member(Z, X),  
        member(Z, Y)).
```

```
?- disjoint([1,2],[3,2,4]).  
no
```

```
?- disjoint([1,2],[3,7,4]).  
yes
```

50 Prolog Negation Problems

```
man(john). man(adam).  
woman(sue). woman(eve).  
married(adam, eve).
```

```
married(X) :- married(X, _).
```

```

married(X) :- married(_, X).
human(X) :- man(X).
human(X) :- woman(X).

% Who is not married?
?- not married(X).
   false

% Who is not dead?
?- not dead(X).
   true

```

51 Prolog Negation Problems

```

man(john). man(adam).
woman(sue). woman(eve).
married(adam, eve).
married(X) :- married(X, _).
married(X) :- married(_, X).
human(X) :- man(X).
human(X) :- woman(X).

% Who is not married?
?- human(X), not married(X).
   X = john ; X = sue
% Who is not dead?
?- man(X), not dead(X).
   X = john ; X = adam ;

```

52 Prolog Negation Problems

- If G terminates then so does not G.
- If G does not terminate then not G may or may not terminate.

```

married(abraham, sarah).

married(X, Y) :- married(Y, X).

?- not married(abraham,sarah).
   false
?- not married(sarah,abraham).
   non-termination

```

53 Open World Assumption

We can program the *open world assumption*:

- A query is either *true*, *false*, or *unknown*.

- A false facts F has to be stated explicitly, using `false(F)`.
- If we can't prove that a statement is *true* or *false*, it's *unknown*.

```
% Philip is Charles' father.
father(philip, charles).
```

```
% Charles has no children.
false(father(charles, X)).
```

54 Open World Assumption

```
prove(P) :- call(P), write('** true'), nl, !.
```

```
prove(P) :- false(P), write('** false'), nl, !.
```

```
prove(P) :-
    not(P), not(false(P)),
    write('*** unknown'), nl, !.
```

55 Open World Assumption

```
father(philip, charles).
false(father(charles, X)).
```

```
% Is Philip the father of ann?
?- prove(father(philip, ann)).
** unknown
```

```
% Does Philip have any children?
?- prove(father(philip, X)).
** true
X = charles
```

```
% Is Charles the father of Mary?
?- prove(father(charles, mary)).
** false
```